

Capítulo 16

Equações diferenciais de segunda ordem

Redução de y'' a duas equações de primeira ordem

Nas seções anteriores, aprendemos métodos da resolução de equações diferenciais da primeira ordem. Também podemos estender estes métodos à *sistemas* de primeira ordem. Resulta que estes métodos podem ser usados também para equações de ordem superior a um, tal como $d^2x/dt^2 = f(t, x(t), x'(t))$, pois esta equação pode ser reduzida a um sistema de duas equações da primeira ordem $y' = f(t, x, y)$ e $x' = y$. Por exemplo: A equação para um pêndulo $x'' + \text{sen}(x) = 0$, $x(0) = 0$, $x'(0) = 1$ pode ser transformada em $x' = y$ e $y' = x'' = -\text{sen}(x)$. Aqui vem mais dois exemplos.

Exemplo 1:

Transforme a equação de **Van der Pol** $y''(x) - m(1-y^2)y'(x) + y = 0$ num sistema de duas equações diferenciais de primeira ordem. m é um parâmetro maior que zero.

Solução:

Para fazer a transformação, vamos aplicar uma mudança de variáveis:

$$y_1(x) = y(x) \quad \text{e} \quad y_2(x) = y'(x).$$

Teremos então o sistema

$$y_1' = y_2$$

$$y_2' = m(1-y_1^2) y_2 - y_1.$$

Para resolver o problema, usando um dos nossos métodos, podemos tomar $m = 2$ e as condições iniciais $y(0) = 2$ e $y'(0) = 0$.

A equação de Van der Pol é uma equação homogênea, pois o lado direito é zero. No próximo exemplo, consideramos uma equação linear de segunda ordem com uma função $F(t)$ "de excitação" no lado direito, por exemplo $F(t) = F_0 \cos(\omega t)$.

Exemplo 2:

Transforme a equação do oscilador harmônico

$$mx''(t) + rx'(t) + kx(t) = F(t)$$

num sistema composto de duas equações diferenciais de primeira ordem. As condições iniciais são $x(t_0) = x_0$ e $x'(t_0) = u_0$.

Solução:

Outra vez fazemos a mudança $x'(t) = y(t)$ e $x''(t) = y'(t)$ com que a equação de segunda ordem torna-se:

$$x'(t) = y \text{ e } y'(t) = f(t,x,y)$$

com as condições iniciais $x(t_0) = x_0$ e $y(t_0) = y_0$. Por exemplo:

$$x''(t) - 2x'(t) + 2x(t) = e^{2t} \text{sen}(t)$$

Para esta equação, utilizaremos o método de **Runge-Kutta** com

$x'(t) = y(t)$ ($=f(t,x,y)$); y representa a derivada dx/dt (= velocidade)
 $y'(t) = 2y(t) - 2x(t) + e^{2t}\text{sen}(t)$ ($=g(t,x,y)$); y' é a derivada segunda x''
 (=aceleração). $t_0=0$; $x_0 = -0.4$; $y_0 = -0.6$ ($=x'(0) = v(0)$).
 A solução analítica é $x(t) = 0.2 e^{2t} (\text{sent} - 2 \text{cost})$

```
Sub Runge_Kutta2()
  Range("A10:E200").Clear
  t = Cells(1, 2).Value
  x = Cells(2, 2).Value
  y = Cells(3, 2).Value
  h = Cells(4, 2).Value
  imax = Cells(5, 2).Value

  Cells(10, 1).Value = t
  Cells(10, 2).Value = x
  Cells(10, 3).Value = y
  Cells(10, 4).Value = FO(t) 'x_analítico

  For i = 1 To imax Step 1
    v1 = F(t, x, y)
    a1 = G(t, x, y)
    v2 = F(t + h / 2, x + v1 * h / 2, y + a1 * h / 2)
    a2 = G(t + h / 2, x + v1 * h / 2, y + a1 * h / 2)
    v3 = F(t + h / 2, x + v2 * h / 2, y + a2 * h / 2)
    a3 = G(t + h / 2, x + v2 * h / 2, y + a2 * h / 2)
    v4 = F(t + h, x + v3 * h, y + a3 * h)
    a4 = G(t + h, x + v3 * h, y + a3 * h)
    x = x + h * (v1 + 2 * v2 + 2 * v3 + v4) / 6
    y = y + h * (a1 + 2 * a2 + 2 * a3 + a4) / 6
    t = t + h
    Cells(10 + i, 1).Value = t
    Cells(10 + i, 2).Value = x
    Cells(10 + i, 3).Value = y
    Cells(10 + i, 4).Value = FO(t)

  Next i
End Sub

Function F(t, x, y)
  F = y
End Function

Function G(t, x, y)
  G = 2 * y - 2 * x + Exp(2 * t) * Sin(t)
End Function

Function FO(t) 'analítico
  FO = 0.2 * Exp(2 * t) * (Sin(t) - 2 * Cos(t))
End Function
```

1	t0:	0							
2	x0:	-0,4							
3	y0:	-0,6							
4	h:	0,1							
5	imax:	5							
6									
7									
8									
9									
10	t	x	y	x_analítico					
11	0	-0,4	-0,6	-0,4					
12	0,1	-0,461733342	-0,631631242	-0,461732971					
13	0,2	-0,525559883	-0,640148948	-0,525559048					
14	0,3	-0,588601436	-0,613663806	-0,588600046					
15	0,4	-0,646612306	-0,536582029	-0,646610284					
16	0,5	-0,693566655	-0,388738097	-0,693563946					

Vemos que o algoritmo de **RK** para uma equação $y' = f(x,y)$ é facilmente modificado para resolver um *sistema* de dois equações de primeira ordem, é só preciso acrescentar uma segunda função $g(t,x,y)$, ou seja, agora temos $x' = f(t,x,y)$ e $y' = g(t,x,y)$.

Bastará então considerar unicamente sistemas de equações de primeira ordem. Mas, as vezes, será mais útil e simples resolver uma equação de segunda ordem por um procedimento direto, ou seja, sem desdobrar a equação em duas equações da primeira ordem.

Para o algoritmo de uma solução direta, partimos das definições de velocidade v e aceleração a (= derivada segunda):

$$t_{n+1} = t_n + h$$

$$y_{n+1} = y_n + h\langle v \rangle$$

$$v_{n+1} = v_n + h\langle a \rangle$$

$\langle v \rangle$ e $\langle a \rangle$ são valores médios de v e a no intervalo $[t_n, t_{n+1}]$ apropriadamente definidos. No esquema de Runge Kutta temos

$$\langle v \rangle := (v_1 + 2v_2 + 2v_3 + v_4)/6$$

$$\langle a \rangle := (a_1 + 2a_2 + 2a_3 + a_4)/6$$

As derivadas calculam-se usando o seguinte esquema:

$v_1 := v$	$a_1 := f(t,y,v)$
$v_2 := v + a_1 h/2$	$a_2 := f(t+h/2, y + v_1 h/2, v_2)$
$v_3 := v + a_2 h/2$	$a_3 := f(t+h/2, y + v_2 h/2, v_3)$
$v_4 := v + a_3 h$	$a_4 := f(t+h, y + v_3 h, v_4)$

Podemos calcular os valores de y e v também pelas relações

$$y = y + hv + h^2(a_1 + a_2 + a_3)/6$$

$$v = v + h(a_1 + 2a_2 + 2a_3 + a_4)/6$$

No caso de um *sistema* de duas equações diferenciais de segunda ordem

$$x'' = f(t,x,x',y,y') \text{ e } y'' = g(t,x,x',y,y')$$

é só necessário escrever o anterior esquema para as duas variáveis x e y com $u = x'$ e $v = y'$. Veja o seguinte programa:

```
Sub Runge_Kutta2()
'para um sistema com x''=F(t,x,y,u,v) e y''=G(t,x,y,u,v)
Range("A10:E200").Clear
t0 = Cells(1, 2).Value
x0 = Cells(2, 2).Value
y0 = Cells(3, 2).Value
u0 = Cells(4, 2).Value
v0 = Cells(5, 2).Value
h = Cells(6, 2).Value
imax = Cells(7, 2).Value
t = t0: x = x0: u = u0: y = y0: v = v0
Cells(10, 1).Value = t
Cells(10, 2).Value = x
' Cells(10, 3).Value = y
Cells(10, 4).Value = u 'vx
'Cells(10, 5).Value = v 'vy
For i = 1 To imax Step 1
  F1 = F(t, x, y, u, v) ' ax
  'G1 = G(t, x, y, u, v) ' ay
  t = t0 + h / 2
  x = x0 + u * h / 2: u = u0 + F1 * h / 2:
  'y = y0 + v * h / 2: v = v0 + G1 * h / 2
  F2 = F(t, x, y, u, v) ': G2 = G(t, x, y, u, v)
  x = x0 + u * h / 2: u = u0 + F2 * h / 2:
  'y = y0 + v * h / 2: v = v0 + G2 * h / 2
  F3 = F(t, x, y, u, v): 'G3 = G(t, x, y, u, v)
  t = t0 + h
  x = x0 + u * h: u = u0 + F3 * h
  'y = y0 + v * h: v = v0 + G3 * h
  F4 = F(t, x, y, u, v) ': G4 = G(t, x, y, u, v)
  x = x0 + h * u0 + h * h * (F1 + F2 + F3) / 6
  ' y = y0 + h * v0 + h * h * (G1 + G2 + G3) / 6
  u = u0 + h * (F1 + 2 * F2 + 2 * F3 + F4) / 6
  'v = v0 + h * (G1 + 2 * G2 + 2 * G3 + G4) / 6
  Cells(10 + i, 1).Value = t
  Cells(10 + i, 2).Value = x
  'Cells(10 + i, 3).Value = y
  Cells(10 + i, 4).Value = u
  'Cells(10 + i, 5).Value = v
  t0 = t: x0 = x: y0 = y: u0 = u: v0 = v
Next i
End Sub
```

```


Function F(t, x, y, u, v)
r = 0.002: w = 3: k = 0.25: FO = 0.016: m = 0.01
  F = (-r * u - k * x + FO * Cos(w * t)) / m
End Function
'Function G(t, x, y, u, v)
  ' G =
'End Function

```

Sem as instruções marcadas como comentários, podemos usar este programa para uma equação só, por exemplo, para a equação


$$x'' = (-rx' - kx + F_0 \cos(\omega t)) / m$$

do **oscilador harmônico** forçado e amortecido.



A fase transitória faz-se notar até quase 30 s. Depois de 30s temos o regime permanente (estacionário).

Com $F_0 = 0$ e $x_0 = 1$ obtemos uma oscilação livre e amortecida:



Trajétória do planeta Mercúrio

Os *sistemas* de equações diferenciais de segunda ordem surgem em muitos problemas da Física, por exemplo, no caso do movimento de um planeta. Antes de calcular a trajetória de tal sistema, temos que falar sobre a redução das variáveis reais a tais sem unidades. Com estas *variáveis reduzidas* podemos utilizar números pequenos e confortáveis para o cálculo numérico.

A única força que atua sobre o planeta é a força gravitacional. As componentes cartesianas desta força são $F_x = m\ddot{x} = -Cm x/r^3$ e $F_y = m\ddot{y} = -Cmy/r^3$. Para a aceleração obtemos as seguintes equações:

$$\ddot{x} = -C x / r^3 \text{ e } \ddot{y} = -C y / r^3 \quad (1)$$

onde significam $\ddot{x} = d^2x/dt^2$, $C := GM$ e $r = (x^2 + y^2)^{1/2}$

O sistema (1) consta de duas equações diferenciais acopladas. (Não tomamos em conta as interações com outras planetas!)

Para simplificar a escrita e os cálculos, introduzimos *novas unidades* para o comprimento e o tempo, ou seja x_0 e t_0 , cujos valores devemos ainda fixar.

Escrevemos $x = X \cdot x_0$, $r = R \cdot x_0$ e $t = T \cdot t_0$. As novas variáveis X , R e T não têm unidades.

A velocidade $v = dx/dt$ toma a forma $v = dx/dt = x_0/t_0 \cdot dX/dT$ e a aceleração é

$$a = d^2x/dt^2 = x_0/t_0^2 \cdot d^2X/dT^2. \quad V = dX/dT = v t_0/x_0.$$

A nova forma da equação $\ddot{x} = -C x / r^3$ será

$$d^2x/dt^2 = x_0/t_0 \cdot d^2X/dT^2 = -C/x_0^2 \cdot X/R^3 \text{ ou seja}$$

$$d^2X/dT^2 = -Ct_0^2/x_0^3 \cdot X/R^3 \quad (2)$$

Só precisamos pôr $Ct_0^2/x_0^3 := 1$ (3), para obter a equação do movimento sem constantes

$$d^2X/dT^2 = -X/R^3 \quad (4)$$

Já que queremos traçar a órbita do Mercúrio, é razoável tomar x_0 igual ao raio da órbita da Terra, que é chamado *unidade astronômica* (A.u.)

$$x_0 = 1,496 \cdot 10^{11} \text{m} \quad (5)$$

Nossa nova unidade do tempo será, então,

$$t_0 = (x_0^3/C)^{1/2} = 5,027 \cdot 10^6 \text{ s} \quad (6)$$

O período do Mercúrio (duração de um ano) é de 88 dias terrestres. Os dados do perihélio são $v_0 = 58,9 \text{ km/s}$ e $r_0 = 46,0 \cdot 10^6 \text{ km}$. Suponhamos que o planeta esteja no perihélio no tempo $T = 0$. Um intervalo de tempo de $\Delta T = 0.05$ significa um tempo real de $\Delta t = \Delta T \cdot t_0 = 0.05 \cdot t_0 = 0.05 \cdot 5,027 \cdot 10^6 \text{ s} = 2.91$ dias. Uma órbita completa corresponderá a 88 dias.

As condições iniciais são:

$$X(0) = x(0)/x_0 = 46 \cdot 10^9 \text{ m}/x_0 = 0,3075$$

$$Y(0) = 0,$$

$$dX(0)/dT = 0,$$

$$dY(0)/dT = v_0 t_0 / x_0 = 58,9 \cdot 10^3 \text{ m/s} \cdot t_0 / x_0 = 1,982$$

Segue aqui o programa:


```
Sub Runge_Kutta2()
'para um sistema com x''=F(t,x,y,u,v) e y''=G(t,x,y,u,v)
Range("A10:E200").Clear
t0 = Cells(1, 2).Value
x0 = Cells(2, 2).Value
y0 = Cells(3, 2).Value
u0 = Cells(4, 2).Value
v0 = Cells(5, 2).Value
h = Cells(6, 2).Value
imax = Cells(7, 2).Value
t = t0: x = x0: u = u0: y = y0: v = v0
Cells(10, 1).Value = t
Cells(10, 2).Value = x
Cells(10, 3).Value = y
Cells(10, 4).Value = u 'vx
Cells(10, 5).Value = v 'vy
For i = 1 To imax Step 1
    F1 = F(t, x, y, u, v) ' ax
    G1 = G(t, x, y, u, v) ' ay
    t = t0 + h / 2
    x = x0 + u * h / 2: u = u0 + F1 * h / 2:
    y = y0 + v * h / 2: v = v0 + G1 * h / 2
    F2 = F(t, x, y, u, v): G2 = G(t, x, y, u, v)
    x = x0 + u * h / 2: u = u0 + F2 * h / 2:
    y = y0 + v * h / 2: v = v0 + G2 * h / 2
    F3 = F(t, x, y, u, v): G3 = G(t, x, y, u, v)
    t = t0 + h
    x = x0 + u * h: u = u0 + F3 * h
    y = y0 + v * h: v = v0 + G3 * h
    F4 = F(t, x, y, u, v): G4 = G(t, x, y, u, v)
    x = x0 + h * u0 + h * h * (F1 + F2 + F3) / 6
    y = y0 + h * v0 + h * h * (G1 + G2 + G3) / 6
    u = u0 + h * (F1 + 2 * F2 + 2 * F3 + F4) / 6
    v = v0 + h * (G1 + 2 * G2 + 2 * G3 + G4) / 6
    Cells(10 + i, 1).Value = t
    Cells(10 + i, 2).Value = x
    Cells(10 + i, 3).Value = y
    Cells(10 + i, 4).Value = u
    Cells(10 + i, 5).Value = v
    t0 = t: x0 = x: y0 = y: u0 = u: v0 = v
Next i
End Sub
```

As funções F e G (= acelerações reduzidas) são

```
Function F(t, x, y, u, v)
  r = (x ^ 2 + y ^ 2) ^ 0.5
  F = -x / r ^ 3
End Function

Function G(t, x, y, u, v)
  r = (x ^ 2 + y ^ 2) ^ 0.5
  G = -y / r ^ 3
End Function
```

Segue a parte inicial dos resultados junto com a órbita:



A órbita se fecha depois de $\approx 1,52$ unidades de tempo, ou seja, depois de ≈ 88 dias. $u = dx/dt$ e $v = dy/dt$.

Seguramente será de interesse uma comparação com um programa profissional como, por exemplo, o MUPAD.

Os resultados que represento em seguida mostram uma divergência a partir da terceira casa decimal. Pode-se supor que o MUPAD trabalha com um algoritmo mais exato do que o de Runge e Kutta. Efetivamente, obtemos melhor concordância com os resultados de MUPAD quando reduzimos os passos de $h = 0,05$ a $0,02$.

Todos os métodos até agora discutidos são chamados de "single-step" (passo único). Isso quer dizer: Quando se conhece a solução $x(t)$ para um instante t determinado, se pode, então, calcular $x(t+h)$, sem necessidade de conhecer também valores da solução para instantes anteriores a t .

Mas, nos chamados métodos "multi-step" (passo múltiplo) faz-se também uso de valores anteriores a t , a saber: $x(t-h)$, $x(t-2h)$, ... Tais métodos precisam, no

começo, de um método "single-step" para calcular alguns valores iniciais para arrancar o algoritmo.

Os *métodos de passo múltiplo* mais populares provêm de ADAMS-BASHFORD, MILNE e de HAMMING. O do ADAMS foi desenvolvido em 1855, baseando-se em idéias do BASHFORD. Anos depois, o método caiu no olvido até, no começo do século XX, foi redescoberto pelo matemático norueguês STRÖMER.

A fórmula de recorrência de ADAMS para $x'(t) = f(x(t))$ tem a seguinte forma:

$$x(t+h) = x(t) + h/24 \cdot (55 f(x(t)) - f(x(t-h)) + 37 f(x(t-2h)) - 9 f(x(t-3h)))$$

Antes de aplicar esta fórmula, calcula-se os valores necessários para o arranque pelo método de Runge-Kutta.

O método de HAMMING é muito exato e estável e é, por isso, usado com frequência.

Programa de MUPAD para calcular a trajetória do Mercúrio:

```

reset();//trajetória de Mercúrio
DIGITS:=5:
x0:=0.3075:y0:=0://posição inicial
vx0:=0:// coordenada-x de v0
vy0:=1.982://coordenada-y de v0
r3(t):=(x(t)^2+y(t)^2)^(3/2):
//Sistema das equações diferenciais com valores iniciais
IVP:={x'(t)=-x(t)/r3(t),y'(t)=-y(t)/r3(t),
x(0)=x0,x'(0)=vx0,
y(0)=y0,y'(0)=vy0}:
fields:=[x(t),y(t),x'(t),y'(t)]:
ivp:=numeric::ode2vectorfield(IVP, fields):
Y := numeric::odesolve2(ivp):

//tabela de valores reduzidos:
print(Unquoted, "T", "X", "Y");
for i from 0 to 2 step 0.2 do
print(i,Y(i)[1],Y(i)[2]):
end_for;

T, X, Y
0, 0.3075, 0.0
0.2, 0.12838, 0.31995
0.4, -0.18895, 0.3647
0.6, -0.41127, 0.1992
0.8, -0.46533, -0.051887
1.0, -0.34053, -0.2822
1.2, -0.065048, -0.37944
1.4, 0.23615, -0.2194
1.6, 0.27459, 0.15302
1.8, 0.000073557, 0.37143
2.0, -0.29572, 0.31621

```

Espalhamento de partículas Alfa

Com o nosso programa do parágrafo anterior é fácil mostrar uma trajetória repulsiva, por exemplo, a trajetória de uma partícula alfa desviada pelo núcleo de um átomo de ouro. Trata-se duma trajetória hiperbólica de repulsão.

A força que atua sobre a partícula Alfa é $F = \frac{1}{4\pi\epsilon_0} \frac{Q_1 Q_2}{r^2}$. Introduzimos uma

constante C pela relação $C = \frac{Q_1 Q_2}{4\pi\epsilon_0 m} = 5,486 \frac{m^3}{s^2}$. $m = 6,65E-27\text{kg}$ é a massa


da partícula Alfa com a carga $Q_1 = 2e$. O núcleo de ouro tem a carga $Q_2 = 79e$.

As duas acelerações são $\ddot{x} = C \cdot x r^{-3}$ e $\ddot{y} = C \cdot y r^{-3}$.


A fim de calcular com números pequenos, precisamos, primeiro, ter uma idéia sobre o tamanho dos passos. O diâmetro do núcleo é aproximadamente $10F = 1E-14$ m. ($1F = 1$ Fermi é usado na física nuclear e tem o valor de $1 \cdot 10^{-15}$ m). Se queremos traçar a trajetória da partícula Alfa sobre $400F$, precisamos $2E-20$ s, supondo uma velocidade de $2E7\text{m/s}$. Se queremos uma trajetória com 50 pontos, precisamos de passos com $h = 4E-22$ s. A partícula poderá partir em $x = 200F = 2E-13\text{m}$. Para o parâmetro de impacto, y_0 , podemos usar valores entre $0,5F$ e $5F$.

As constantes têm, neste exemplo, os seguintes valores reduzidos: $x_0 = 5$, $y_0=1$, $u_0 = -0,8$, $h = 0,15$ e $C = 1$.

O valor de y_0 é o parâmetro de impacto b. Na figura usamos $y_0 = b = 1F$.



Aumentando a distancia do impacto a $y_0 = 2$ faz com que a partícula não seja retro espalhado, ela somente sofrerá um desvio:



Movimento num campo r^{-1}

Até agora consideramos campos centrais inversamente proporcionais ao quadrado da distância -e a natureza vigia estritamente sobre a preservação do expoente 2. As digressões de este valor são menores de $2 \cdot 10^{-16}$. Mas, nos laboratórios, podemos realizar casos com expoentes bem diferentes de 2, por exemplo o expoente 1 num filtro eletrostático para as velocidades de partículas carregadas, como elétrons.

Neste parágrafo, utilizamos, outra vez, o programa do planeta Mercúrio, mas, esta vez, para o caso de uma força inversamente proporcional à distância, $F(r) = k/r$.

Injeta-se um elétron perpendicularmente num campo elétrico que se forma em torno de um fio reto infinito e carregado uniformemente com q Coulomb por metro, C/m. O elétron vai descrever trajetórias em torno do fio. Um caso especial será uma órbita circular.

Por simetria, as linhas de força são radiais e, se q for positiva, dirigidas para fora do fio. Pela aplicação da lei de Gauss podemos facilmente mostrar que

$$E(r) = q/(2\pi\epsilon_0) \cdot r^{-1}$$

A constante ϵ_0 é denominada *permissividade elétrica do vácuo*. No SI de unidades seu valor é $8,854 \cdot 10^{-12} \text{ C}^2 \text{ N}^{-1} \text{ m}^{-2}$.

Para a força que atua sobre o elétron obtemos

$$F(r) = -qe/(2\pi\epsilon_0) \cdot r^{-1}$$

As duas componentes cartesianas da aceleração são

$$x'' = -C \cdot x \cdot r^{-2} \quad \text{e} \quad y'' = -C \cdot y \cdot r^{-2}$$

onde $C = q e / (2\pi\epsilon_0 m_e)$ e $r = (x^2 + y^2)^{1/2}$

(Compare com $x'' = C \cdot x \cdot r^{-3}$; $y'' = C \cdot y \cdot r^{-3}$ do parágrafo anterior !)

e = carga elementar: $1,602177 \cdot 10^{-19}$ C

m_e = massa do elétron: $9,10939 \cdot 10^{-31}$ kg

e/m_e = razão carga/massa para o elétron: $1,7588 \cdot 10^{11}$ C/kg

Para que o elétron se mova sobre uma *órbita circular*, deve ser $F(r) = -mv^2/r$. Desta relação resulta a seguinte equação para a velocidade v

$$v = \sqrt{\frac{qe}{2\pi\epsilon_0 m_e}} = \sqrt{C}$$


Ou seja, elétrons que se movem ao longo de uma trajetória circular têm a mesma velocidade, independente do raio.

Primeiro vamos usar variáveis reduzidas, depois colocamos as variáveis com as unidades normais de s , m e m/s .

No primeiro programa escolhemos como novas unidades $x_0 = 10^{-2}$ m e $t_0 = 10^{-6}$ s. Uma unidade natural para a velocidade seria $v_0 = 10^4$ m/s. Esta escolha tem como consequência que $C = 10^8$ m² s⁻². Para que C tenha este valor, temos que tomar $q = 3,163 \cdot 10^{-14}$ C/m. Utilizando estes valores, obtemos **$Ct_0^2/x_0^2 = 1$** .


Assim, no primeiro programa, $y_0 = 5$ significa um comprimento de $5 \cdot 10^{-2}$ m = 5 cm. A velocidade $u_0 = 1$ significa uma velocidade real de 10^4 m/s. O intervalo $h = 0.5$ é, na realidade, igual a $5 \cdot 10^{-7}$ s.

```
Function F(t, x, y, u, v)
  r = (x ^ 2 + y ^ 2): K = 1
  F = -K * x / r
End Function
Function G(t, x, y, u, v)
  r = (x ^ 2 + y ^ 2): K = 1
  G = -K * y / r
End Function
```




Vemos a trajetória circular para $v = 10^4$ m/s. Você poderá utilizar outros raios, ou seja, outros valores para y_0 , p.ex. $y_0 = 1$, para ver que sempre obterá um círculo como trajetória.

Se você mudar a velocidade de $u_0 = 1$ para $u_0 = 0,5$ (= 5 000 m/s), o resultado será uma trajetória em forma de roseta que não se fechará. Veja a seguinte figura:



Agora fazemos o mesmo cálculo usando variáveis normais. No programa introduzimos $y_0 = 0.05\text{m}$, $u_0 = 1\text{E}4\text{m/s}$ e $h = 5\text{E}-8\text{s}$. Para a constante C obtemos $C = 1,000\text{E}8$.



O gráfico é, outra vez, um círculo como na figura anterior, mas, nos eixos temos metros em vez de centímetros.

O átomo hidrogênico

O átomo de hidrogênio é constituído por um elétron e um próton. Devido à sua simplicidade, o átomo de hidrogênio desempenhou um papel central no desenvolvimento da física quântica.

Apesar de ser um sistema simples, o problema de resolver a equação de Schrödinger para o átomo de hidrogênio é bem complicado, pois trata-se de um problema tridimensional, onde a função U depende da coordenada radial r .

Para a grande maioria dos problemas existentes na natureza, as respectivas equações de Schrödinger não podem ser resolvidas exatamente, e o átomo de hidrogênio e sua série isoeletrônica (He^+ , Li^{++} , etc.) pertencem a este pequeno grupo dos problemas exatamente resolvíveis. O H^+ é um átomo de hélio ionizado, o Li^{++} é um átomo de lítio duplamente ionizado.

Segue aqui a equação radial de Schrödinger:

$$\frac{1}{r^2} \frac{d}{dr} \left(r^2 \frac{dR}{dr} \right) + \frac{2\mu}{\hbar^2} [E - U(r)] R = l(l+1) \frac{R}{r^2} \quad (1)$$

No caso do átomo de hidrogênio temos, porém, uma dificuldade adicional, pois a equação (1), que queremos resolver, tem dois termos com singularidades, em $1/r$ e $1/r^2$. No programa, começamos os cálculos não em $r = 0$, mas sim em $r + d$, onde d é um número pequeno, $1E-8$, que nos protege do perigo de uma divisão por zero.

Primeiro será preciso de formular a equação (1) numa forma mais apropriada para o cálculo numérico.

Vamos mudar a variável r por a variável sem dimensão ρ , definida por

$$\rho := 2Z/na_0 \cdot r := \alpha \cdot r \quad (2)$$

O número quântico, n , é definido por

$$E := -\mu e^4 Z^2 / (4\pi\epsilon_0)^2 \hbar^2 \cdot n^{-2} \quad (3)$$

$a_0 = \epsilon_0 \hbar^2 / \pi \mu e^2$ ou $a_0 = \hbar^2 / \mu e^2$ (cgs) $\approx 0,529 \cdot 10^{-8} \text{cm}$ é o primeiro raio de Bohr. Para expressar E no sistema cgs é preciso substituir ϵ_0 por $1/4\pi$: $E_{\text{cgs}} = -\mu e^4 Z^2 / (2\hbar^2 n^2)$.

Aplicando estas abreviaturas, podemos reduzir a equação de Schrödinger a uma forma bastante simples

$$R''(\rho) + 2/\rho R'(\rho) + (n/\rho - 1/4 - l(l+1)/\rho^2) R(\rho) = 0 \quad (4)$$

Usamos $R(\rho)$ em vez de $R(r)$ para indicar que estamos usando a variável adimensional $\rho = \alpha r$ com $\alpha = 2Z/na_0$.

O programa a seguir utiliza o método de Runge-Kutta com os valores de contorno $R(0) = 0$ e $R'(0) = 0.2041$. Na proximidade de $\rho = 0$ empregamos uma aproximação linear: $f = x + u \cdot t$.

Agora, consideramos o caso com $n=3$ e $l=0$. Para usar a variável ρ , temos que substituir r/a_0 por $3\rho/2$. Isso significa que $e^{-r/3a_0} = e^{-\rho/2}$.


O programa calcula a *densidade de probabilidade radial* $R(r)^2 r^2$. Esta função nos dá a probabilidade de encontrar o elétron num átomo de hidrogênio em certa distância r do núcleo (próton)

Apresentamos aqui por fines de comparação a função analítica da distribuição radial.

$$(n=3, l=0): R(r) = 2(1/3a_0)^{3/2} (1 - 2/3 r/a_0 + 2/27 (r/a_0)^2) e^{-r/3a_0}$$

$$\text{ou } R(\rho) = a_0^{-3/2} / 9\sqrt{3} \cdot (6 - 6\rho + \rho^2) e^{-\rho/2}$$

Na planilha a seguir, comparamos os valores numéricos com a solução analítica.



```

Sub Runge_Kutta2() ' x''(t,x,x')
Range("A10:D5000").Clear
t0 = Cells(1, 2).Value
x0 = Cells(2, 2).Value
u0 = Cells(3, 2).Value
h = Cells(4, 2).Value
imax = Cells(5, 2).Value
t = t0: x = x0: u = u0
Cells(10, 1).Value = 0
Cells(10, 2).Value = (x0 * t0 * 1.5) ^ 2
Cells(10, 3).Value = (F0(0) * t * 1.5) ^ 2

For i = 1 To imax Step 1

    F1 = F(t, x, u) ' ax
    t = t0 + h / 2
    x = x0 + u * h / 2: u = u0 + F1 * h / 2:
    F2 = F(t, x, u)
    x = x0 + u * h / 2: u = u0 + F2 * h / 2:
    F3 = F(t, x, u):
    t = t0 + h
    x = x0 + u * h: u = u0 + F3 * h
    F4 = F(t, x, u)
    x = x0 + h * u0 + h * h * (F1 + F2 + F3) / 6
    u = u0 + h * (F1 + 2 * F2 + 2 * F3 + F4) / 6
    t0 = t: x0 = x: u0 = u
    m = 10
    If i Mod (m) = 0 Then ' dirige a frequência de impressão
        Cells(10 + i / m, 1).Value = t * 3 / 2
        Cells(10 + i / m, 2).Value = (x * t * 1.5) ^ 2
        Cells(10 + i / m, 3).Value = (F0(t) * t * 1.5) ^ 2
    End If

Next i
End Sub

```


No programa, utilizamos a função $F = -2*u/t + (1/4 + l*(l+1)/t^2 - n/t)*x$ apenas a partir de $t > 0,00001$.

```
Function F(t, x, u)
  n = 3: l = 0
  If t > 0 And t <= 0.00001 Then F = x + u * t Else
  If t > 0.00001 Then
    F = -2 * u / t + (1 / 4 + l * (l + 1) / t ^ 2 - n / t) * x
  End If
End Function

Function FO(t)
  FO = 1 / (9 * 3 ^ 0.5) * (6 - 6 * t + t ^ 2) * Exp(-t / 2)
End Function
```

O gráfico do estado (3,0) tem três picos. O valor mais provável fica na distância $r \approx 13,5 \cdot a_0 \approx 13,5 \cdot 0,529 \cdot 10^{-8} \text{cm} = 7,14 \cdot 10^{-8} \text{cm}$. A teoria dá para a distância mais provável (valor esperado) a expressão

$$\langle r_{nl} \rangle = \frac{n^2 a_0}{Z} \left[1 + \frac{1}{2} \left(1 - \frac{l(l+1)}{n^2} \right) \right] \quad (5)$$

Em nosso caso, obtemos desta equação também $\langle r_{30} \rangle = 9a_0(1+0.5) = 13.5a_0$.

Uma representação detalhada da teoria do átomo de hidrogênio pode-se encontrar no site <http://www.geocities.com/Athens/Agora/6594/indexsci.html>